

## Un disjoncteur électronique réarmable

J'invite le lecteur à consulter le site pour des informations complémentaires.

Page d'accueil du site Internet :

[page d'accueil](#)

d'autres pdf, sur différents sujets :

[liste des PDF](#)


Cet article montre le schéma d'une électronique de protection placée entre une alimentation et une carte électronique. C'est un montage qui surveille le courant, et, en cas de surcharge ou de court-circuit (provoqué par l'utilisateur qui « bricole » sous tension), coupe l'alimentation de la carte, tel un disjoncteur. C'est à l'utilisateur (auteur de la maladresse...) de réarmer l'installation par une action manuelle.

### 1) Analyse du schéma.

#### 1.a) Description et principe de fonctionnement

Le montage utilise 4 transistors et un bouton poussoir extérieur.

L'alimentation est fonctionnelle et fournit un potentiel en **A**. Le disjoncteur s'intercale entre l'alimentation et la carte en service. Posons  $I$  le courant qui circule dans  $R_{shunt}$  et  $Q_4$  (on néglige les courants de base). Et appelons  $I_{seuil}$  le courant maximum avant disjonction.



Les numéros de nœuds sont en rouge.

$Q_1, Q_4$  :  
2N2907A

$Q_2, Q_3$  :  
2N2222

$R_{shunt}$  : 10  $\Omega$

$R_2$  : 10 k $\Omega$

$R_3$  : 1 k $\Omega$

$R_4$  : 1 k $\Omega$

$R_5$  : 10 k $\Omega$

$R_6$  : 10 k $\Omega$

$R_s$  : 100  $\Omega$

Alimentation 5 V.

- En marche normale ( $I < I_{seuil}$ ),  $Q_4$  est conducteur, en régime saturé. La tension qui se présente à l'entrée de la carte sous tension est :  $V(S) = V(A) - R_{shunt} I - V_{CEsat4}$ .

Pour avoir l'état saturé de  $Q_4$ , il faut un courant de base tel que  $I_{B4} \gg I_{C4}/\beta$ .

C'est la maille  $R_3, R_4, Q_3$  qui assure cette condition.  $Q_3$  est en mode normal ( $V_{BE}$  direct,  $V_{BC}$  inverse) par  $R_5, R_6$ .  $Q_1$  est ouvert, et  $Q_2$  également.

- Si la carte vient à consommer un courant supérieur à  $I_{seuil}$ , la tension aux bornes de  $R_{shunt}$  dépasse 0,6 V et le transistor  $Q_1$  entre en conduction. Un courant émetteur collecteur de  $Q_1$  s'établit et entre dans la base de  $Q_2$ . Ce dernier est alors polarisé en mode normal, et son courant de base est tel qu'il passe en saturé. Le potentiel **B3** tombe alors à  $V_{CEsat2}$ , ce qui est inférieur à 0,6 V, bloquant ainsi  $Q_3$ . Dans ces conditions, il n'y a plus de courant de base dans  $Q_4$  qui se bloque. La carte n'est alors plus alimentée. La protection a eu lieu.

Sans courant  $I$ , la tension aux bornes de  $R_{shunt}$  tombe à 0 V, ce qui bloque  $Q_1$ , et par conséquent  $Q_2$ . On est alors dans un état stable, les 4 transistors sont coupés.

- C'est l'action du bouton poussoir qui permet de sortir de cette situation (on suppose que l'évènement qui a provoqué la surcharge n'est plus) : un appui permet de replacer une tension en **S**, et donc en **B3**. Le transistor  $Q_3$  peut donc repasser en mode conducteur, tirant alors un courant de base de  $Q_4$ .  $Q_4$  repasse en mode saturé, et le bouton poussoir n'a plus besoin d'être appuyé. On est de nouveau dans la situation standard :  $Q_3, Q_4$  ON,  $Q_1, Q_2$  OFF.


La simulation de ce montage nécessite quelques explications.

### 1.b) Comment simuler une résistance variable

Pour pouvoir simuler ce circuit, il nous faut faire varier la consommation de la carte, pour représenter ce que serait un défaut de surcharge. Sur le schéma, le courant  $I$  est établi par  $R_L$ , variable.

Sous PSpice, on peut réaliser une **résistance commandée par une tension** grâce à une « astuce » :


En simulation, une résistance peut être remplacée par un quadripôle formé d'une tension commandée par une autre tension. Si cette dernière est l'image de  $I$ , il suffit de lui affecter un coefficient de valeur  $R$  pour avoir en sortie l'équation  $U = R I$ , qui est la loi d'Ohm.



Sous Pspice, le captage de courant est fait simplement par une source de tension nulle placée en série. Pour plus de facilité, la résistance variable ainsi formée est intégrée dans un **sous-circuit** que l'on nomme **XRvar**.

Pour avoir une résistance commandée par une tension  $V_{com}$ , il suffit de placer l'équation de sortie :  $R I (1 + \alpha V_{com})$ .

Le paramètre  $\alpha$  permet d'avoir un nombre d'ohms pour un nombre de volts.



Vérifions le fonctionnement de cette « résistance », au travers un schéma de test :




Schéma de test, netliste ci-contre.

```

résistance commandée
* fichier test_rvar.cir
Valim U 0 DC=30V
R1 U S 150 ; en série avec RW
XRvar S 0 com 0 Rvar
Vcom com 0 PWL (0,0 10,0 30,5 40,5
+ 40.1,0 60,0 60.1,5 60.2,5 60.3,0)
.param R0 = 150
.param alpha = -0.2

.SUBCKT Rvar Rout1 Rout2 in1 in2 ;
E Rout1 i
+ VALUE = { {R0}*I(V) * (1+{alpha}*(V(in1,in2)) ) }
V i Rout2 DC 0
.ENDS
.TRAN 10m 70 0 10m
.probe
.end
    
```

On fixe arbitrairement une résistance série  $R_1 = 150 \Omega$  et une alimentation de 30 V.

On choisit alors le coefficient  $R_0 = 150$ . L'équation est alors  $150 I (1 + \alpha V_{com})$ .

Si  $V_{com} = 0$ , la résistance entre S et 0 vaut  $150 \Omega$ . Valim débite dans  $300 \Omega$ , et le courant est de 0,1 A.

Si  $V_{com}$  augmente, la résistance diminue, car  $\alpha$  est négatif.

Le choix d'avoir  $\alpha = -0,2$  donne la valeur de -1 au terme  $(\alpha V_{com})$  si  $V_{com} = 5$  V. Dans ce cas, l'équation aboutit à une résistance nulle. L'alimentation débite dans  $R_1$ , et le courant vaut 0,2 A.

On va simuler avec une tension  $V_{com}$  qui évolue entre 0 V et 5 V.

### Simulation

La commande évolue de 0 V à 5 V entre t = 10 s et t = 30 s.

De 30 s à 40 s, on laisse 5 V.

Puis on chute instantanément.


On provoque ensuite une impulsion à t = 60 s

L'affichage du rapport tension / courant donne la résistance : 150 Ω par défaut (Vcom nul), puis décroissance vers un court-circuit, conformément à la loi de commande.

Le courant circulant dans la résistance commandé suit la loi d'Ohm :  $I = V(U) / (R_1 + XRvar)$ .

Sa valeur fluctue entre 100 mA et 200 mA.

En S, on a le pont diviseur de tension :  $V(S) = V(U) R_1 / (R_1 + XRvar)$



Résistance commandée par une tension

### Interprétation :

De 0 à 10 s, la tension de commande vaut 0 V. La résistance XRvar vaut 150 Ω. Comme elle est en série avec  $R_1 = 150 \Omega$ , on a un courant de 100 mA. Le potentiel en S est alors  $V_{lim} / 2$ , par pont diviseur.

De 10 s à 30 s, la tension de commande évolue linéairement de 0 V à 5 V. La résistance XRvar décroît linéairement de 150 Ω à 0 Ω. Le courant monte jusqu'à 200 mA, comme attendu. Le pont diviseur évoluant, le potentiel en S décroît, jusqu'à 0 V, car XRvar présente un court-circuit.


On laisse cette situation jusqu'à 40 s, instant quand on remet instantanément Vcom à 0 V. On se retrouve comme à t = 0.

Puis, à 60 s, on simule un court-circuit fugitif, par une brève impulsion à 5 V. Le courant dans la maille fait donc une impulsion de 200 mA.

En conclusion, par cet artifice, un changement continu de tension provoque un changement continu de résistance  $R_L$ .

### 1.c) Comment simuler un bouton poussoir

Il existe sous Pspice l'interrupteur commandé : le composant « Voltage-Controlled Switch », qui est une résistance commandée en tension. Sa valeur est entre  $R_{ON}$  et  $R_{OFF}$ , respectivement quand on applique  $V_{ON}$   $V_{OFF}$  sur une tension de commande.




En interne, le passage de  $R_{ON}$  à  $R_{OFF}$  suit une loi complexe (même très complexe !), mais monotone.

Pour réaliser un bouton poussoir, on applique  $V_{ON}$  (d'où court-circuit) un court instant et  $V_{OFF}$  (d'où résistance très grande) le reste du temps.

On choisit ici  $R_{ON} = 100 \Omega$ , (ce qui permet d'avoir  $R_s$  pris en compte),  $R_{OFF} = 10 \text{ M}\Omega$ . Le signal de commande est arbitrairement de 1 V, et le passage ON OFF est à 0,5 V.

### 2) Simulation du disjoncteur.

Remarque : La durée des évènements est fixée de façon arbitraire. Pour donner un certain réalisme à une manipulation vraie, on raisonne à l'échelle des secondes et de la minute.



#### 2.a) test du dépassement de seuil et du court-circuit

Dans cette simulation, l'utilisateur actionne le bouton poussoir :

- une première fois à  $t = 5 \text{ s}$ ,
- et une seconde fois à  $t = 50 \text{ s}$




La commande évolue de 0 V à 5 V entre  $t = 10 \text{ s}$  et  $t = 30 \text{ s}$ .

Cela signifie que la carte présente, durant ces 20 secondes une résistance apparente qui évolue entre  $150 \Omega$  et  $0 \Omega$ .

Le court-circuit est maintenu durant 10 s, puis est coupé.


à  $t = 60 \text{ s}$ , un court-circuit fugitif a lieu.




Le courant absorbé par la carte est de 30,6 mA par défaut, et croit quand la résistance diminue.

A la valeur 55,2 mA, la sécurité intervient, et maintient le courant sensiblement constant, puis chute à 0.

L'action du bouton poussoir à  $t = 50 \text{ s}$  réalimente la carte.



Au début, la carte est alimentée par 4,6 V, et quand la sécurité intervient, son alimentation chute pour tomber à 0 V : le disjoncteur a coupé la carte.



Après remise sous tension à 50 s, le court-circuit bref fait disjoncter

fonctionnement

## Interprétation :

- De 0 à 10 s, la tension de commande vaut 0 V. La résistance XRvar vaut 150  $\Omega$  : elle représente la charge demandée par la carte. Par la présence des chutes de tension dans R<sub>shunt</sub> (348 mV) et V<sub>CEQ4</sub> (62 mV), il n'y a que 4,59 V sur la carte. Le courant consommé par la carte est de 30,6 mA.

Bien qu'inutile, on fait le test du bouton poussoir à 5 s, (simulé ici par une impulsion de 1 V sur le commande du Switch). Cela ne modifie quasiment rien sur les potentiels ni courant, car le bouton poussoir court-circuite une tension quasi nulle.

- Pour simuler une augmentation de courant consommé par la carte, on diminue la résistance apparente de XRvar en augmentant Vcom. Cela se produit de 10 s à 30 s (c'est la même commande que le test précédent de XRvar seul). A 30 ms, XRvar présente 0  $\Omega$  : on a été jusqu'au court-circuit.

On voit clairement la croissance du courant. A la valeur 55,2 mA, Q<sub>1</sub> se met à conduire, mettant ainsi la protection en service. Le courant I est limité à 55,2 mA, et, comme la résistance apparente continue de diminuer, la tension en S chute linéairement. Quand V(S) s'approche de 1,5 V, V(B3) diminue jusqu'au blocage de Q<sub>3</sub>, coupant ainsi Q<sub>1</sub> : la tension en S tombe alors à 0.

- La protection ayant eu lieu, on simule maintenant le retour à une situation plus normale : à t = 40 s, Vcom retombe à 0 V, ce qui est équivalent, en pratique, par le retrait du court-circuit par le « bricoleur » qui a retiré le défaut qui était sur la carte. Il faut donc réarmer le disjoncteur.

Cela se fait ici à t = 50 s : on ferme fugitivement le switch. Instantanément la tension de sortie repasse à 4,59 V, et le courant revient à la valeur créée par 150  $\Omega$ .

- La simulation pourrait en rester là, mais on continue le test en provoquant, volontairement un bref court-circuit à 60 s : en simulation, il suffit de placer fugitivement la valeur de 5 V en Vcom, et la résistance de XRvar présente 0  $\Omega$  un court instant. Cela modélise une manipulation hasardeuse sous tension comme celle de faire toucher les fils qui amènent la tension et la masse... On voit très nettement que le courant a tendance à monter, mais retombe à 0 par l'action de la protection.

## 2.b) Que se passe-t-il si le court-circuit est permanent ?

Quitte à analyser le comportement du disjoncteur face à une maladresse, continuons l'expérience : à son insu, l'utilisateur a provoqué un court-circuit franc et durable sur son alimentation de la carte. Comme cela a disjoncté, il appuie sur le bouton poussoir. Et...

- la tension en S reste à 0 V. C'est évident, car c'est un court-circuit franc et durable...

- Comme Q<sub>3</sub> est bloqué, Q<sub>4</sub> ne peut pas devenir conducteur. Donc le disjoncteur ne peut pas réamorcer.

Remarque : l'association série Rshunt et R<sub>s</sub> (= 10 + 100 = 110  $\Omega$ ) est soumise à V(A) = 5 V.

D'où I(Rshunt) = 45 mA. Sans la présence de R<sub>s</sub>, ce courant serait excessif pour Q<sub>1</sub> qui aurait une tension V<sub>EB</sub> destructrice.

En conclusion, un court-circuit fugitif actionne le disjoncteur qui peut être réarmé. Un court-circuit permanent est sans danger pour l'alimentation.

## 2.c) Que se passe-t-il à la mise sous tension ?

- quand l'utilisateur met sous tension V(A), le disjoncteur peut ne pas entrer en conduction naturellement, faute de courant dans Q<sub>4</sub>, car V(S) = 0. (Une solution serait d'ajouter une résistance R<sub>7</sub> entre B4 et la masse, afin d'avoir Q<sub>4</sub> conducteur, même légèrement : dès que Q<sub>4</sub> entre en conduction, il s'auto alimente par Q<sub>3</sub> et va en saturation. Sauf que cela entraîne une consommation de courant interne quand c'est disjoncté, ce qui est un point négatif).

*→ A cause des conditions initiales,  
la carte peut ne pas être alimentée après la mise sous tension*

- Sur les cartes électroniques, il est d'usage de placer, en parallèle des alimentations, des condensateurs de découplage. Ceux-ci sont fréquemment de 2 technologies différentes : électrolytiques aluminium, de forte capacité, pour faire « capa réservoir », et de type céramique, de faible capacité, mais adapté aux hautes fréquences. L'association parallèle des 2 types permet de pallier l'impédance interne non nulle des alimentations.

Mais cela occasionne un appel de courant lors de la mise sous tension de la carte, pour charger ces condensateurs. Et qui dit appel de courant, peut signifier pointe de courant supérieur à Iseuil, et donc, déclenchement de la disjonction...

Cette pointe de courant est très liée à la technologie de l'alimentation (forme d'onde de l'établissement de la tension s'établit, impédance interne) et à celle des condensateurs (présence de ESR, résistance série, ou d'une inductance série).

*(Pour information, sous l'article 78, paragraphe 7, on peut observer l'établissement de la tension 5 V d'une alimentation de laboratoire).*

*→ A cause de la présence des condensateurs de découplage, la protection peut disjoncter lors de la mise sous tension*

Pour éviter ces (éventuels) défauts, l'utilisation du disjoncteur présenté dans cet article nécessite la procédure suivante : l'opérateur met sous tension son alimentation, puis amorce le disjoncteur par le bouton poussoir ; dès lors, la carte est alimentée et la protection est active.

## Conclusion

Le montage présenté est simple, facile à comprendre, et les simulations montrent le fonctionnement. On peut modifier le seuil de déclenchement en changeant Rshunt : en plaçant  $1 \Omega$  (soit 10 fois moins), le courant sera environ 550 mA (soit 10 fois plus) pour provoquer la même chute de tension de 0,6 V. Bien entendu, il faut que les transistors Q<sub>1</sub> et Q<sub>4</sub> soit adaptés à ces courants.

Si l'utilisateur désire une tension de sortie de 5 V pour alimenter sa carte, il faut utiliser une alimentation réglable, ajustée à 5,4 V.

## Annexe 1 : fichier disjoncteur.cir, prêt à simuler

```
fusible réarmable
* fichier disjoncteur.cir
*
.lib eval.lib

.Model int Vswitch (Ron=100 Roff=10Meg Von=0.55 Voff=0.45) ; interrupteur

* circuit :

Valim A 0 DC=5
Rshunt A E4 10
Q1 C1 E4 A Q2N2907A ; C B E
R2 C1 B2 10k

Q2 B3 B2 0 Q2N2222

Vbp bp 0 PWL (0,0 4.9,0 5,1 5.1,1 5.2,0 50,0 50.1,1 50.2,1 50.3,0)

Sbp E4 S bp 0 int ; int int com com bp d'amorçage

Q4 S B4 E4 Q2N2907A ; C B E
R3 E4 B4 1k

R4 B4 C3 1k
Q3 C3 B3 0 Q2N2222 ; CBE

R5 S B3 10k
R6 B3 0 10k

* utilisateur : résistance variable
XRvar S 0 Com 0 RW
Vcom com 0 PWL (0,0 10,0 30,5 40,5 40.1,0 60,0 60.1,5 60.2,5 60.3,0)

.param R0 = 150 ; 30 mA sous 5 V
.param alpha = -0.2 ; pente

.SUBCKT RW Rout1 Rout2 in1 in2 ;
E Rout1 i VALUE = { I(V)*{R0}*(1+{alpha}*(V(in1,in2))) }
V i Rout2 DC 0
.ENDS

.OP
.TRAN 10m 70 0 10m

.probe
.end
```

articles 1 à 43 : sur le livre

**Tableau récapitulatif des articles PDF disponibles sur ce site**

n°	titre	lien présentation	lien direct article
	Guide d'installation et d'emploi simplifié	<a href="#">présentation</a>	<a href="#">document PDF</a>
44	Exemples basiques et des exercices...	<a href="#">présentation</a>	<a href="#">document PDF</a>
45	Un exemple de circuit passif	<a href="#">présentation</a>	<a href="#">document PDF</a>
46	Un oscillateur Colpitts	<a href="#">présentation</a>	<a href="#">document PDF</a>
47	Compensation en fréquence des amplificateurs opérationnels	<a href="#">présentation</a>	<a href="#">document PDF</a>
48	Un amplificateur à transistors bipolaires	<a href="#">présentation</a>	<a href="#">document PDF</a>
49	Une bascule D Flip Flop CMOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
50	Une porte XOR à transistors MOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
51	Un VCO à 12 transistors MOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
52	Une PLL à moins de 20 transistors MOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
53	Un oscillateur à résistance négative	<a href="#">présentation</a>	<a href="#">document PDF</a>
54	Une charge électronique	<a href="#">présentation</a>	<a href="#">document PDF</a>
55	Un amplificateur en classe C	<a href="#">présentation</a>	<a href="#">document PDF</a>
56	Le monostable 74 123	<a href="#">présentation</a>	<a href="#">document PDF</a>
57	Un amplificateur en classe D	<a href="#">présentation</a>	<a href="#">document PDF</a>
58	Le transformateur en linéaire	<a href="#">présentation</a>	<a href="#">document PDF</a>
59	La loi d'ohm thermique	<a href="#">présentation</a>	<a href="#">document PDF</a>
60	Le transformateur en non linéaire	<a href="#">présentation</a>	<a href="#">document PDF</a>
61	Robustesse d'un oscillateur en anneau	<a href="#">présentation</a>	<a href="#">document PDF</a>
62	Une alimentation stabilisée	<a href="#">présentation</a>	<a href="#">document PDF</a>
63	Modélisation d'un haut-parleur	<a href="#">présentation</a>	<a href="#">document PDF</a>
64	Un synthétiseur de fréquence	<a href="#">présentation</a>	<a href="#">document PDF</a>
65	Un ampli audio de Sparkfun	<a href="#">présentation</a>	<a href="#">document PDF</a>
66	Simulation logique et analogique	<a href="#">présentation</a>	<a href="#">document PDF</a>
67	Un oscillateur à relaxation	<a href="#">présentation</a>	<a href="#">document PDF</a>
68	Lecteur de TAG RFID 125 kHz	<a href="#">présentation</a>	<a href="#">document PDF</a>
69	Diagramme de l'œil avec Pspice	<a href="#">présentation</a>	<a href="#">document PDF</a>
70	Un amplificateur hautes fréquences	<a href="#">présentation</a>	<a href="#">document PDF</a>
71	Une bizarrerie enfin expliquée...	<a href="#">présentation</a>	<a href="#">document PDF</a>
72	Comprendre le paramétrage de la FFT	<a href="#">présentation</a>	<a href="#">document PDF</a>
73	La relation de Bennett	<a href="#">présentation</a>	<a href="#">document PDF</a>
74	Simuler un circuit à plus de 20 transistors avec PSpice Eval	<a href="#">présentation</a>	<a href="#">document PDF</a>
75	Une horloge biphase sans recouvrement	<a href="#">présentation</a>	<a href="#">document PDF</a>
76	Quelques simulations sur la diode	<a href="#">présentation</a>	<a href="#">document PDF</a>
77	Un ampli classe A, avec transformateur de sortie	<a href="#">présentation</a>	<a href="#">document PDF</a>
78	Des stimuli pour PSpice	<a href="#">présentation</a>	<a href="#">document PDF</a>
79	Simuler le TL431 : zener ajustable	<a href="#">présentation</a>	<a href="#">document PDF</a>
80	Un ADC flash	<a href="#">présentation</a>	<a href="#">document PDF</a>
81	Une chaine d'acquisition : S&H, ADC, DAC	<a href="#">présentation</a>	<a href="#">document PDF</a>
82	Un amplificateur 50 MHz	<a href="#">présentation</a>	<a href="#">document PDF</a>
83	Un dérivateur non inverseur	<a href="#">présentation</a>	<a href="#">document PDF</a>
84	Un amplificateur bipolaire avec push pull CMOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
85	Rôle des répéteurs logiques dans un circuit intégré	<a href="#">présentation</a>	<a href="#">document PDF</a>
86	Un driver logique CMOS pour charge 50 ohms	<a href="#">présentation</a>	<a href="#">document PDF</a>
87	Des triggers de Schmitt et des applications	<a href="#">présentation</a>	<a href="#">document PDF</a>
88	Un filtre gaussien analogique	<a href="#">présentation</a>	<a href="#">document PDF</a>
89	Un générateur de bruit rose	<a href="#">présentation</a>	<a href="#">document PDF</a>

90	Un anémomètre à fil chaud : simulation comportementale	<a href="#">présentation</a>	<a href="#">document PDF</a>
91	Un oscillateur à pont de Wien stabilisé par CTN	<a href="#">présentation</a>	<a href="#">document PDF</a>
92	L'emballage thermique d'une diode	<a href="#">présentation</a>	<a href="#">document PDF</a>
93	Les puissances dans un amplificateur	<a href="#">présentation</a>	<a href="#">document PDF</a>
94	Asservissement de puissance dans une résistance	<a href="#">présentation</a>	<a href="#">document PDF</a>
95	Asservissement de la puissance émise par une antenne radio	<a href="#">présentation</a>	<a href="#">document PDF</a>
96	Un driver de LED de puissance	<a href="#">présentation</a>	<a href="#">document PDF</a>
97	Exploiter Pspice pour simuler des filtres numériques	<a href="#">présentation</a>	<a href="#">document PDF</a>
98	Un filtre en cosinus surélevé avec Pspice	<a href="#">présentation</a>	<a href="#">document PDF</a>
99	Effet de la température sur un amplificateur en classe A	<a href="#">présentation</a>	<a href="#">document PDF</a>
100	Un amplificateur à transistors JFET et bipolaires	<a href="#">présentation</a>	<a href="#">document PDF</a>
101	Un disjoncteur électronique réarmable	<a href="#">présentation</a>	<a href="#">document PDF</a>
102	Un redressement à thyristors	<a href="#">présentation</a>	<a href="#">document PDF</a>
<b>Supplément, hors article :</b>			
<b>mon cours « Electronique pour les communications numériques », <a href="#">polycopié couleur 201 pages en pdf</a></b>			

[retour à l'écran d'accueil de ce site](#)